HEWLETT-PACKARD

HP-71/HP-75 to HP 150/IBM PC/XT/AT File Transfers

Summary

This application note provides instructions for transferring ASCII (TEXT) files between an HP-71 or HP-75 Handheld Computer and an HP 150 or IBM PC/XT/AT computer.

Procedures, instructions, and/or programs are provided for each step in the transfer process. These steps include:

- Enter the BASIC data transfer program (XFERPC) into handheld computer (either HP-71 or HP-75)
- Configure the HP 150 or IBM computer
- Connect system components Listings of programs for the following functions are included:
- Interactive configuration of the HP 82164A HP-IL/RS-232C Interface
- TEXT file upload from an HP-71 or HP-75 to either an HP 150 or IBM PC/XT/AT
- TEXT file download from an HP 150 or IBM PC/XT/AT to an HP-71 or HP-75

This note assumes you have a working knowledge of the products involved.

Appendix A provides instructions and programs to allow conversion of HP-71 or HP-75 DATA files into TEXT files. After conversion, these TEXT files can be uploaded to the HP 150 or IBM using the XFERPC program for the appropriate handheld computer.

HP-71 or HP-75 BASIC programs may be transformed into TEXT files for uploading. Refer to Owner's Manual for instructions.

*The programs in this document have not been tested with a modem.

Crosstalk[®] is a registered trademark of Microstuf, Inc.

Indicates an HP 150 or IBM key.

Required Equipment

 Either: HP-71 Handheld Computer and HP 82401A HP-IL Interface
 OR
 HP-75 Handheld Computer plus 00075-15001 I/O ROM or
 Users' Library 00075-13013 I/O

00075-15001 I/O ROM or Users' Library 00075-13013 I/O Utilities

- HP 82164A HP-IL/RS-232C Interface and HP-IL Cables (2)
- Either: HP 150A, B, or C, RS-232C Cable and HP 45431A Advance/Link or HP 45425A DSN/Link OR

IBM PC, AT, or XT, RS-232C Cable and Crosstalk[®] XVI (Data Communications System Software)

Hardware Configuration*

HP-IL/RS-232C Interface

Configure the HP-IL/RS-232C Interface as a DCE terminal. See page 21 of the Interface Owner's Manual for instructions.

RS-232C Cable (for HP 150) Use a male to female RS-232C cable with pins wired end-to-end.

RS-232C Cable (for IBM) Use a cable with pins wired end-to-end, one connector female and the other connector as required by the COM1 port of the IBM.

HP-71/HP-75 to HP 150 File Transfers

This section provides an explanation of how to enter the XFERPC program into your handheld computer; configure the HP 150, PORT 1 and MS-DOS; and connect system components.

Procedure

- 1. Enter the program XFERPC on either the HP-71 or HP-75 (refer to Owner's Manual for instructions). Be sure to use the BASIC program specified for the handheld computer you are using. If you are using the HP-75, the "HPILCMDS" file must be resident in the machine before loading XFERPC. Both the Users' Library I/O Utilities and the HP-75 I/O ROM contain HPILCMDS.
- 2. Configure the HP 150 as follows:
 - Ensure that the HP 150 is in terminal mode. (Press (f6) from the main menu if in P.A.M.)
 - Press User System), f8 (config keys), then f1 (global config).
 - Configure the HP 150 as a computer and ensure that Port 1 is the remote device.
 - Press (1) to save the configuration.
- 3. Configure Port 1 as follows:
 - Press (f8) (config keys), then (f3) (Port 1 config). The following menu will appear.

Indicates an HP-71 or HP-75 key.



WordStar[®] is a registered trademark of MicroPro International.

&MSG '^[&dB Choose another softkey to continue. '

Operation

Transferring files from the HP-71 or HP-75 to the HP 150 (Upload)

This section provides an explanation of the steps required to upload TEXT files from your handheld computer to the HP 150.

- 1. Configure the HP 150 and run Advance/Link (or DSN/Link).
- 2. From the main menu, press (f4) (Logging Options), then (f1) (Local file). Enter the name of the local log file (the file to be stored after uploading) and press (Return). If a file by that name does not already exist, an asterisk will appear in the "Local File" and "Logging" key labels. The HP 150 is now ready to receive data.
- Run "XFERPC" on the HP-71 or HP-75. When prompted, press
 to upload data. (It is not necessary to press ENDLINE / <u>RTN</u>.) Enter the name of the file you wish to upload to the HP 150. "Uploading data . . ." will appear in the display.
- 4. Each line of uploaded data will appear in the display of the HP 150 as it is received. When the transfer is complete, press (f) to store the uploaded file on disc.
- 5. "Transfer another file?" will appear in the display of the HP-71 or HP-75 when the transfer is complete. Press Y to transfer another file, or N to end the program.

Transferring files from the HP 150 to the HP-71 or HP-75 (Download)

This section provides an explanation of the steps required to download files from the HP 150 to your handheld computer.

- 1. Configure the HP 150 and run Advance/Link (or DSN/Link).
- 2. From the main menu, press (f3) (Command), then (f1) (Command File). The HP 150 will prompt for the name of the command file to be used to download files to the handheld computer. Enter "DOWNLOAD", using a disc label identifier if necessary, and press (Return).
- Run "XFERPC" on the HP-71 or HP-75. Press D to download data. (It is not necessary to press ENDLINE / RTN.) Enter the name of the file to be downloaded from the HP 150. This name need not be the same name as the file being downloaded from the HP 150. "Downloading data..." will appear in the display. At this point the HP-71 or HP-75 is ready to receive data.
- 4. On the HP 150, enter the name of the local file to be downloaded, being sure to include a disc label identifier if applicable. Press (Return) to begin the transfer. Once the data has been downloaded, "Transfer Complete" will appear in the HP 150 display.
- 5. "Transfer another file?" will appear in the display of the HP-71 or HP-75 when the transfer is completed. Press Y to transfer another file, or N to end the program.

6. Depending on the format of the HP 150 source file (the file being downloaded), the resulting HP-71 or HP-75 TEXT file may contain unnecessary "blank" lines. If these are unacceptable, they should be deleted.

HP-71/HP-75 to IBM PC/XT/AT File Transfers

This section provides an explanation of how to set up the system; enter the XFERPC program into your handheld computer; and set up script and command files for the Crosstalk software on the IBM.

Procedure

- 1. Set up the system as follows:
 - Connect the RS-232C cable to the HP-IL/RS-232C Interface and to COM1 of the IBM.
 - Connect the HP-IL cables to the HP-IL/RS-232C Interface and to either the HP-71 or HP-75. NOTE: If using the HP-75, the interface must be the first device on the HP-IL loop.
- 2. Enter the program XFERPC on either the HP-71 or HP-75. Be sure to use the BASIC program specified for the handheld computer you are using. If you are using the HP-75, the "HPILCMDS" file must be resident in the machine before loading XFERPC. (Both the Users' Library I/O Utilities and the HP-75 I/O ROM contain this file.)
- 3. If you don't already have one, follow the Crosstalk instructions to create a working copy of the Crosstalk disc. Put this disc in drive B.

With the MS-DOS disc in drive A, use EDLIN to input the XFERPC.XTS Script File (see programs at the end of this note) onto the Crosstalk disc. Where applicable, press (+) and (6) to input the control character "∧".

The IBM will prompt you with: A> On IBM type in: edlin b:xferpc.xts Press: (ENTER)

5. Make B the default drive. The IBM will prompt you with: A>

Type in: b:

Press: (ENTER)

Then load the Crosstalk software. The IBM will prompt you with: B> Type in: xtalk

Press: (ENTER)

6. Choose NEWUSER from the selection displayed on the IBM. Set up the XFERPC.XTK file by responding to the prompts as follows (press (ENTER) after each response unless otherwise noted):

Prompts	Responses
Enter phone number: Enter system name or comments:	RS232 XFERPC
0110,300,600,1200,2400,4800, or 9600 baud?	9600
Do you want to save this setup in a command file?	y (don't press ENTER)
Save settings in what file? To begin the command prompt.	XFERPC.XTK
Press ENTER: Command? (DO NOT type in GO)	(<u>enter</u>) LO XFERPC.XTS

- 7. Follow the instructions in the display on the IBM from this point on. Subsequent use of the routine will load XFERPC.XTS automatically when you load the XFERPC command file.
- 8. Sample prompts and responses for downloading a file named TEST from the IBM PC to the HP-71 or HP-75 are shown below.
 - On the IBM:

Prompts	Response		
Do you want to UPLOAD, DOWNLOAD or END transfer routine?			
 (IBM will display) To DOWNLOAD a file to the HP71 or HP75: ON HP: Run XFERPC, then following instructions in the display, choose D, enter filename. ON IBM: As requested on the command line, press ENTER, then type SEND filename then ENTER again. When COMMAND? returns, transfer is done, type in DO then ENTER 			
Script processing suspended – use DO a press ENTER:	lone to restart.		

9. Implement the above instructions by doing the following:

- On the HP-71 or HP-75, key in: run 'XFERPC' (quotation marks are optional with the HP-71)
- Subsequent HP-71 or HP-75 prompts and your responses are:

Prompts	Responses
Upload or Download? File to Download: ? Downloading data	D TEST (no response needed)
Prompts and responses on the IBM are:	
Prompts	Responses
Command? Sending b:testpress Esc to stop.	(ENTER) send test (no response needed)
Subsequent command lines on the IBM should be similar to the following:	
Prompts	Responses
Command? do	
Do you want to transfer another file? Y or N	Y or N

Operation

Transferring files between the IBM and HP-71 or HP-75 (Upload and Download)

This section provides an explanation of the steps required to upload and download files between the IBM and HP-71 or HP-75.

- 1. The first time you transfer files, follow all the steps described in the Procedure section.
- 2. After the first time you transfer files, follow steps 1, 2 and 5 in the Procedure section. Then:
 - Choose XFERPC from the selection displayed on the IBM.
 - To upload or download files, type in (U) or (D) and follow the instructions in the IBM display.
- 3. You will have the opportunity to end the transfer routine when you are finished uploading and downloading files. The IBM will display: Do you want to transfer another file? Y or N
 - Press Y or N and follow the prompts.

The HP-71 or HP-75 will display: Transfer another file?

- Press Y to transfer another file or N to end the program.
- 4. Depending on the format of the IBM source file (the file being downloaded), the resulting HP-71 or HP-75 TEXT file may contain unnecessary "blank" lines. If these are unacceptable, they should be deleted.

Programming

XFERPC Program for the HP-71 Handheld Computer

# Keystrokes	Comment
100 CALL XFERPC @ END !REV A 8/01/85	
110 SUB XFERPC!	
120 R \$ = PEEK\$('2F946',4)	Save delay setting.
130 DELAY 0,0	
140 DIM L\$[96]	
150 RESTORE IO	
160 REMOTE :RS232	
170 OUTPUT :RS232	
;'SBE;C0;C2;P4,SW0;SS0;SL7;'	Configure HP 82164A using
180 LOCAL	remote mode instructions.
190 DISP 'Upload or Download?'	
200 K $\$ = KEY$ $\$ @$ IF K $\$ = ' '$ THEN 200	
210 IF K\$='U' THEN GOSUB 310	Upload to PC.
220 IF K \$='D' THEN GOSUB 460	Download from PC.
230 IF K\$<>'U' AND K\$<>'D' THEN BEEP	
@ GOTO 190	
240 DISP 'Transfer another file?'	
250 K\$=KEY\$ @ IF K\$=' ' THEN 250	
260 IF K\$='Y' THEN 190	Transfer another file.
270 IF K \$ = 'N' THEN 680	End program.
280 BEEP @ GOTO 240	
290 OFF ERROR	Error handling routine.
300 DISP F\$;' not found' @ BEEP @ WAIT 1	Nonexistent file name.
310 INPUT 'File to Upload: ','?';F\$	Upload routine.
320 IF F\$='?' THEN 310	
330 ON ERROR GOTO 290	
340 CAT F\$	Insure specified file exists.
350 OFF ERROR	1
360 DISP 'Uploading data'	
370 ASSIGN #1 TO F\$	
380 ON ERROR GOTO 430	Prepare for end-of-file
	condition.
390 READ #1:L\$	Read a line of text.
400 OFF ERROR	
410 OUTPUT :RS232 :L\$	

Li	ne		

Keystrokes

Comment

420 GOTO 380 430 OFF ERROR 440 IF ERRN<>54 THEN BEEP @ DISP If not end of file, error halts 'ERR L';ERRL;':'; ERRM\$ @ PAUSE program. 450 GOTO 630 460 INPUT 'File to Download: ','?';F\$ 470 IF F\$='?' THEN 460 Download routine. 480 DISP 'Downloading data...' 490 CREATE TEXT F\$ 500 ASSIGN #1 TO F\$ 510 DIM L1\$[96] 520 F = FLAG(-23, 1)Save Flag - 23 setting and set Flag – 23. 530 L\$=' ' 540 OUTPUT :RS232 USING '#,K';CHR\$(17) Signal PC to send a line of text. 550 ENTER :RS232 ;L1\$ Receive data from PC. 560 L\$=L\$&L1\$ Add data to current line. 570 IF NOT POS(L\$,CHR\$(13)) THEN 550 Check for end of record. 580 IF L\$[1,1] = CHR\$(4) THEN 620 Check for end of file. 590 PRINT #1;L\$[1,POS(L\$,CHR\$(13))-1] Store a line of text. 600 L=L[POS(L, CHR(13)) + 1]Update current line. 610 IF POS(L\$,CHR\$(13)) THEN 580 ELSE 540 Check for another end-ofrecord. 620 F = FLAG(-23, F)Restore Flag - 23 setting. 630 ASSIGN #1 TO * 640 DISP 'Transfer complete' 650 WAIT 1 660 RETURN 670 DISP 'Done' 680 POKE '2F946',R\$ Restore delay setting. 690 END SUB 700 SUB VER(V\$)! Version subprogram. 710 V\$='A' @ END SUB

# Keystrokes	Comment
100 ASSIGN IO ':RS'	4
110 ! REV A 8/01/85	
120 DELAY 0	
130 DIM L\$[96]	Configure HP 82164A
':RS'.'UNL.REN.LAD#'.'SBE:C0:C2:P4:SW0:	using Remote mode
SS0;SL7;'	instructions
50 SENDIO ' ','NRE',' '	
60 DISP 'Upload or Download?'	
170 K = UPRC(KEY) @ IF K = ' ' THEN 170	
80 IF K $=$ 0 THEN GOSUB 280 90 IE K $=$ 10' THEN COSUB 430	Upload to PC.
200 IF K = D IHEN GOSOD 450	Download noniti C.
(a) GOTO 160	
210 DISP 'Transfer another file?'	
220 K\$=UPRC\$(KEY\$) @ IF K\$=' ' THEN 220	
230 IF K \$ = 'Y' THEN 160	Transfer another file.
240 IF K \$ = 'N' THEN 610	End program.
250 BEEP @ GOTO 210	Envoy han dling nouting
200 OFF EKKOK 270 DISP ES: ' not found' @ BEEP @ WAIT 1	Nonevistent file name
280 INPUT 'File to Upload: '.'?':F\$	Upload routine.
290 IF $F_{5} = '?'$ THEN 280	
300 ON ERROR GOTO 260	
310 CAT F\$	Insure specified file exists.
320 OFF ERROR	
330 DISP Uploading data	
350 ON FRROR GOTO 400	Prepare for end of file
	condition.
360 READ # 1 ; L\$	Read a line of text.
370 OFF ERROR	
380 SENDIO	
(10) (10) (10) (10)	Send a line of text to PC.
100 OFF FRROR	
410 IF ERRN<>34 THEN BEEP @ DISP	If not end of file, error halts
'ERROR line';ERRL @ STOP	program.
420 GOTO 570	
130 INPUT 'File to Download: ','?';F\$	Download routine.
$\begin{array}{llllllllllllllllllllllllllllllllllll$	
460 ASSIGN # 1 TO F\$ TFXT	
470 DIM L1\$[96]	
1480 L = ' '	
490 SENDIO ':RS','UNL,LAD#',CHR\$(17)	Signal PC to send a line of
	text.
500 L1 = ENTIO(':RS','UNL, TAD#, SDA')	Receive data from PC.
520 IF NOT POS(I \$ CHR\$(13)) THEN 500	Check for end of record
530 IF L \$[1,1] = CHR\$(4) THEN 570	Check for end of file.
540 PRINT # 1 ; L\$[1,POS(L\$,CHR\$(13))–1]	Store a line of text.
550 L \$ = L\$[POS(L\$,CHR\$(13)) + 1]	Update current line.
560 IF POS(L\$,CHR\$(13)) THEN 530 ELSE 490	Check for another end of
TO ASSIGN # 1 TO *	record.
580 DISP 'Transfer Complete'	
590 WAIT 1	
600 RETURN	
610 DISP 'Done'	

XFERPC Program for the HP-75 Handheld Computer

620 END

XFERPC.XTS Script File for Crosstalk Transfer Program.

Line

Keystrokes

1 LABEL Begin 2 GOLOCAL 3 screen s 4 cls 5 message 6 7 Do you want to UPLOAD, DOWNLOAD or END transfer routine? 8 9 ASK Enter U, D, or E: If -UDE Jump Begin 10 JUMP DO-@ 11 12 13 LABEL DO-U 14 screen s 15 cls 16 message 17 18 To UPLOAD a file from the HP71 or the HP75: ON IBM: As requested on Command line press ENTER, then type CA 19 20 filename and press ENTER again. 21 ON HP: Run XFERPC choosing U. Follow the instructions in the 22 display. ON IBM: When transfer is done, type in DO and press ENTER. 23 24 25 wait MA 26 CA -27 cls 28 ASK Do you want to transfer another file? Y or N 29 If Y rwind 30 JUMP DO-E 31 LABEL DO-D 32 33 LW Char "^Q" 34 screen s 35 cls 36 message 37 38 To DOWNLOAD a file to the HP71 or the HP75: 39 ON HP: Run XFERPC, then following instructions in the display, choose D, 40 enter filename. 41 ON IBM: As requested on the command line, press ENTER, then type SEND 42 filename then ENTER again. When COMMAND? returns, transfer 43 is done, type in DO then ENTER. 44 45 wait MA REPLY "^M" Press "Crtl M" 46 REPLY "^D^M" Press "Crtl D Crtl M" 47 48 cls 49 ASK Do you want to transfer another file? Y or N 50 If Y rwind 51 JUMP DO-E 52 53 LABEL DO-E 54 cls ASK Do you want to leave XTALK? Y or ENTER 55 56 If Y QUit 57 DO-58 LO-59 ; the end

Appendix A

Converting DATA Files to TEXT Files in Preparation for Transfer

The following program, in separate versions for the HP-71 and HP-75, can be used to convert a DATA file into a TEXT file, thus extending the functionality of the XFERPC program. This program is analogous to a hypothetical BASIC command "TRANSFORM <data file> INTO TEXT", which is not available on the HP-71 or HP-75. (Actually, the transformation of an HP-75 BASIC file containing DATA statements into TEXT is possible, but the resulting TEXT file probably won't be in the format desired for data transfer.)

The program is called DATATEXT (convert DATA file to TEXT) and works independently of the program called XFERPC. To use it in conjunction with XFERPC, run DATATEXT first to convert one or more DATA files to TEXT, then use XFERPC for the actual data transfer.

Because of the differences between DATA and TEXT files, DATATEXT must use an arbitrary method for conversion. The method is defined as follows:

- Each numeric item in a DATA file is converted to a string of the minimum number of numerals consistent with presenting maximum accuracy (this corresponds to the HP-75 display format and the HP-71 format STD). The numeral string is then stored sequentially into the TEXT file.
- Each string item in a DATA file is stored, unchanged, sequentially into the TEXT file. String items longer than 96 characters cause an error, and must be modified before using DATATEXT. (Refer to HP-71 or HP-75 Owner's Manual for details.)

Procedure

- 1. Load the correct version of DATATEXT into the HP-71 or HP-75. Note that during the conversion process, DATATEXT requires that there be enough free RAM for both the original DATA file (if in main memory) and the resulting TEXT file.
- Type: RUN "DATATEXT" <u>ENDLINE</u> / <u>RTN</u>. (Quotation marks are optional with the HP-71.)
- 3. At the prompt, "DATA file: ?", key in the name of the DATA file to be converted to TEXT and press ENDLINE / RTN. With an HP-75, the DATA file must be in main memory; with an HP-71, a DATA file on mass storage or independent RAM can be indicated by appending a device specifier to the file name.
- The computer will display "* Working *" as each DATA file item is converted to text and stored into a dummy TEXT file called "XXX".
 - When completed, the computer will purge the original DATA file, so be sure to keep a copy if needed.
 - Next, the computer will rename or copy the dummy file "XXX" to the original DATA file name, purge "XXX" (if applicable), and then display "Done".

Programming DATATEXT Program for the HP-71

Lin	Line		
#	Keystrokes	Comment	
100	CALL DATATEXT	-	
110	END		
120	SUB DATATEXT!		
130	R\$ = PEEK\$('2F946',4)	Save DELAY setting.	
140	DELAY 0		
150	D\$ = PEEK\$('2F6D9',5)	Save display format.	
160	STD		
170	DIM S\$[96]		
180	INPUT 'DATA file: ','?';F\$		
190	DISP TAB(6); ^{**} Working *'		
200	CREATE TEXT XXX	Create dummy TEXT file.	
210	ASSIGN #1 TO F\$		
220	ASSIGN #2 TO XXX		
230	ON ERROR GOTO 310	Set up "data type" and end	
240		of file detection.	
240	KEAD #1;N	Attempt to read numeric	
250	OFE EPROP	item.	
250	$S_{\pm}^{\pm} = STR_{\pm}^{\pm}(N)$	Convert numericitor to	
200	$3\phi = 31 \text{K}\phi(\text{IN})$	convert numeric item to	
270	COTO 200	string.	
280	READ #1.55	Read string item	
290	PRINT #2:5\$	Store string item or string-	
270	110111 // 2,00	formatted numeric item	
300	GOTO 230	formatica numeric item.	
310	OFF ERROR	Error handling routine.	
320	IF ERRN = 31 THEN 280	If error is "data type", read	
		string item.	
330	IF ERRN<>54 THEN BEEP @ DISP 'ERR	If not end of file, error halts	
	L';ERRL;':';ERRM\$ @ PAUSE	program.	
340	ASSIGN #1 TO *	1 0	
350	ASSIGN #2 TO *		
360	PURGE F\$		
370	COPY XXX TO F\$	Copy new file to old file	
		name.	
380	PURGE XXX		
390	POKE '2F6D9',D\$	Restore display format.	
400	DISP 'Done'		
410	POKE '2F946',R\$	Restore DELAY setting.	
420	END SUB		

Program on next page

DATATEXT Program for the HP-75

Line #	Keystrokes	Comment
100	DELAY 0	
110	DIM S\$[96]	
120	INPUT 'DATA file: ','?';F\$	
130	DISP TAB(12);'* Working *'	
140	ASSIGN #1 TO F\$	
150	ASSIGN #2 TO "XXX",TEXT	Create dummy TEXT file.
160	ON ERROR GOTO 210	Set up end of file detection.
170	READ #1 ; 5\$	Read string item or numeric
180	OFF ERROR	
190	PRINT #2 ; S\$	Save string item or string- formatted numeric item.
200	GOTO 160	
210	OFFERROR	Error handling routine.
220	IF ERRN<>34 THEN BEEP @ DISP 'ERROR	If not end of file, error halts
	line';ERRL @ STOP	program.
230	ASSIGN #1 TO *	
240	ASSIGN #2 TO *	
250	PURGE F\$	
260	RENAME 'XXX' TO F\$	Copy new file to old file name.
270	DISP 'Done'	
280	END	

For additional information about Hewlett-Packard's handheld computers in manufacturing solutions, call the HP office nearest you and ask for your technical computer or instrument representative. Local HP sales offices are listed in the white pages of your telephone book.

Hewlett-Packard Handheld Computer and Calculator Operation 1000 N.E. Circle Blvd. Corvallis, Oregon 97330



Printed in USA 10/85 5954-1318 WCP 4.0